

Projekt 3 *Der Makrorekorder*

Die Programmiersprache VBA stellt in den einzelnen Applikationen mehrere Tausend Elemente zur Verfügung, darunter fallen Anweisungen wie MsgBox, Objekte, Eigenschaften, Methoden, Konstanten und vieles mehr. Sie haben im vorherigen Projekt gelernt, dass es ohne Hilfe nicht geht: Selbst wenn Sie die Anweisung oder das Element kennen, brauchen Sie Unterstützung, um die Syntax richtig benutzen oder die Anzahl der Argumente bestimmen zu können.

Ein wertvolles, für den Einsteiger unverzichtbares Programmierinstrument ist der Makrorekorder. Er zeichnet alle Aktionen auf, die Sie an der Oberfläche Ihrer Applikation durchführen, und wandelt diese in sofort ausführbare Prozeduranweisungen um. Die Suche nach dem richtigen Befehl für eine Aktion bleibt Ihnen damit zumindest für die Bearbeitung von Dokumenten, Mappen, Tabellenblättern und Zellen sowie die Anwendung von Menüs und Symbolen erspart.

Lernen Sie in diesem Projekt, wie Sie mit Hilfe des Makrorekorders schneller zu Resultaten kommen. Diese Lektionen erwarten Sie:

- 1** Tabelle vorbereiten
- 2** Makrorekorder starten
- 3** Makro aufzeichnen
- 4** Aufzeichnung beenden
- 5** Makroaufzeichnung bearbeiten
- 6** Aufgezeichnete Makros verbinden
- 7** Makro für Rechnungsblatt schreiben

Lektion

Tabelle vorbereiten

Das nächste Makro, wieder in der Applikation Excel erstellt, soll folgende Aufgaben übernehmen: Der Benutzer gibt den Namen eines Rechnungsempfängers in eine Eingabebox ein. Das Makro sucht in einer Adresskartei nach diesem Namen und überträgt die komplette Adresse in ein Rechnungsblatt.

1

Erstellen Sie im Excel-Fenster ein Tabellenblatt mit Adressen.

	A	B	C	D	E	F	G
1	Firma	Anrede	Vorname	Name	Straße	PLZ	Ort
2	Herberts GmbH	Herrn	Daber	Herberts	Goethestr. 30	50311	Köln
3	ARCOS AG	Frau	Beate	Widemann	Hoffnerplatz 7	75290	Fädenhausen
4	SysTex GmbH	Herrn	Theo	Ingelmann	Kempinskistr. 58	10321	Berlin
5	Fröhlich & Sohn	Herrn	Gundner	Schwapp	Heidemännstraße 34a	76345	Weierstadt
6	TBT Verlag	Frau	Senta	Christians	Engelschallingerstr. 4	80912	München
7	Kunz & Partner	Frau	Olga	Ludwigs	Erlangerstr. 34a	92322	Nürnberg
8	Braun Damenmoden	Frau	Lieselotte	Braun	Riederweg 45	79888	Weinberge
9	Rosenberg GmbH	Herrn	Rudolf	Rosenberg	Hafnergasse 6	71080	Stuttgart
10							
11							
12							
13							
14							
15							
16							
17							
18							
19							
20							
21							
22							

TIPP

Am schnellsten geht's per Doppelklick auf den Tabellennamen. Sie können aber auch Format, Blatt, Umbenennen wählen.

2

Weisen Sie dem Blatt den Blattnamen »Kundenkartei« zu.

16							
17							
18							
19							
20							
21							
22							

Geben Sie dem Bereich den Bereichsnamen »Adressen«.


Den Bereich markieren Sie am schnellsten mit **Strg** + **↑** + ***** (der Zellzeiger muss im Bereich stehen). Tragen Sie den Bereichsnamen links oben in das Namensfeld ein oder wählen Sie *Einfügen, Namen, Definieren*.

Lektion

Makrorekorder starten

Speichern Sie Ihre Mappe zur Sicherheit noch einmal ab und starten Sie dann die Aufzeichnung der geplanten Aktionen.

Wechseln Sie in das zweite Tabellenblatt und starten Sie den Makrorekorder.




Der Wechsel ist wichtig, damit der Rekorder auch die Aktivierung der richtigen Tabelle aufzeichnet.

2

Geben Sie den Namen des neuen Makros ein – hier *RechnungNeu*.


ACHTUNG

Makronamen sollten Sie immer ohne Leerzeichen schreiben!



3

Entscheiden Sie sich für diese Arbeitsmappe als Speicherort.




Die persönliche Arbeitsmappe wäre eine globale Mappe, die andere Mappen steuert. Sie heißt PERSONL.XLS und befindet sich im Startordner XLSTART.

4

Tragen Sie eine Beschreibung ein und starten Sie die Aufzeichnung.

Beschreibung




3 Lektion

Makro aufzeichnen

Wenn der Makrorekorder aktiv ist, werden alle Aktivitäten aufgezeichnet. Hüten Sie sich also vor Maus- und Tastatureingaben, die Sie nicht brauchen, sie produzieren unnötigen VBA-Code. Verschieben Sie auch keine Symbolleisten und arbeiten Sie nicht mit Rollbalken.

1

Zeichnen Sie den ersten Schritt auf: Wechseln Sie in die Adresskartei.




Symbol Aufzeichnung beenden

Statuszeilenhinweis: Aufzeichnung läuft

Die Aufzeichnung läuft, führen Sie nur Aktionen durch, die Sie aufzeichnen wollen (Symbolleiste nicht verschieben!).


2

Zeichnen Sie den zweiten Schritt auf: Markieren Sie die erste Adressenspalte durch einen Klick auf den Spaltenkopf.



3


Zeichnen Sie den dritten Schritt auf: Suchen Sie nach einem Namen.



Starten Sie die Suchaktion mit dem Befehl *Suchen* aus dem Bearbeiten-Menü.

4


Geben Sie einen beliebigen Firmennamen ein und bestätigen Sie mit **OK**.



5

Beenden Sie die Suchaktion.

Das Suchergebnis wird markiert



4

Lektion

Aufzeichnung beenden


Der gesuchte Firmenname ist markiert, der Zellzeiger steht auf dem Suchergebnis. Das reicht uns für das Rechnungs-Makro. Damit wissen Sie, in welcher Zeile die Adresse steht.

1

Beenden Sie die Aufzeichnung.

ACHTUNG

Nicht die Symbolleiste schließen, sondern das erste Symbol anklicken!




Die Symbolleiste wird geschlossen, aus der Statuszeile verschwindet der Kommentar »Aufzeich.«.

2


Schalten Sie in das VBA-Editor-Fenster um.

TIPP


Wenn die Symbolleiste nicht sichtbar ist, können Sie die Aufzeichnung auch mit Extras, Makro, Aufzeichnung beenden abschließen.




Wechseln Sie in das neue Modulblatt.



Das Ergebnis sehen Sie, wenn Sie auf das neue Modulblatt umschalten, das der Makrorekorder automatisch angelegt hatte: Die Anweisungsfolge steht in Form von VBA-Befehlen im Modul.



Der Makrorekorder hat alle Aktionen in passende VBA-Befehle umgesetzt. Für die Suchen-Aktion verwendet er das Objekt Selection (die Markierung im Tabellenblatt) und die Methode Find. Alle Einstellungen in der Dialogbox Suchen werden im Makrobefehl in Form von Argumenten festgehalten.



5 Lektion

Makroaufzeichnung bearbeiten

Nun würde diese Aufzeichnung natürlich immer wieder nach der gleichen Adresse suchen. Ändern Sie den Code soweit um, dass der Benutzer des Makros den Nachnamen als Suchbegriff eingeben kann.

1

Fügen Sie gleich nach der Sub-Anweisung eine Eingabeaufforderung ein und dimensionieren Sie die Variable.

TIPP

Dim-Anweisungen schreiben Sie immer unmittelbar nach der Sub-Anweisung.

```
Option Explicit
Sub RechnungNeu()
Dim suchname
' RechnungNeu Makro
' Makro sucht nach einem Firmennamen in der Adresskartei.
Sheets("Kundenkartei").Select
Range("A1").Select
Columns("A:A").Select
suchname = InputBox("Bitte geben Sie die Firma ein:", _
    "Adresse suchen")
```

2

Prüfen Sie die Abbrechen-Schaltfläche ab.

ACHTUNG

Verwenden Sie keine Anführungszeichen für die Variable!

```
Option Explicit
Sub RechnungNeu()
Dim suchname
' RechnungNeu Makro
' Makro sucht nach einem Firmennamen in der Adresskartei.
Sheets("Kundenkartei").Select
Range("A1").Select
Columns("A:A").Select
suchname = InputBox("Bitte geben Sie die Firma ein:", _
    "Adresse suchen")
If suchname = "" Then Exit Sub
```

Mit der IF-Anweisung prüfen Sie, ob der Benutzer der Variable etwas übergeben hat. Falls nicht, wird das Makro kommentarlos beendet.

3

Weisen Sie der Find-Anweisung die Variable für den Suchnamen zu.


```
Option Explicit
Sub RechnungNeu()
Dim suchname
' RechnungNeu Makro
' Makro sucht nach einem Firmennamen in der Adresskartei.
Sheets("Kundenkartei").Select
Range("A1").Select
Columns("A:A").Select
suchname = InputBox("Bitte geben Sie die Firma ein:", _
    "Adresse suchen")
If suchname = "" Then Exit Sub
Selection.Find(What:=suchname, _
    After:=ActiveCell, _
    LookIn:=xlFormulas, _
    LookAt:=xlPart, _
    SearchOrder:=xlByRows, _
    SearchDirection:=xlNext, _
    MatchCase:=False).Activate

End Sub
```

An Stelle des aufgezeichneten Namens fügen Sie jetzt einfach die Variable ein, die von der InputBox mit dem Suchbegriff übergeben wird.

4

Testen Sie das geänderte Makro.



Starten Sie das Makro gleich aus dem Editor heraus mit **[F5]**. Das Tabellenblatt wird aktiviert, und die InputBox-Anweisung blendet ein Dialogfenster ein. Geben Sie einen Suchnamen ein, das Makro wird diesen in der Spalte A markieren.


Lektion 6

Aufgezeichnete Makros verbinden

Mit dem Suchergebnis, das Sie übrigens auch schon durch Eingabe eines Namens- teils (z.B. »Her« für »Herberts«) erhalten, wird die Zelle markiert, die den gesuchten Nachnamen enthält. Zeichnen Sie ein weiteres Makro auf, das eine neue Tabelle erstellt, und binden Sie den Code des ersten Makros mit dem zweiten zusammen.

1

Zeichnen Sie die Aktion für ein neues Tabellenblatt auf.



Makrorekorder ist aktiv

Starten Sie den Makrorekorder, den Namen *Makro2* können Sie unverändert übernehmen.

2

Wählen Sie *Einfügen, Tabellenblatt*.

3


Beenden Sie die Aufzeichnung.

4

Schalten Sie in das Aufzeichnungsmodul um.


TIPP

Die Aufzeichnung beenden Sie wie zuvor mit dem Befehl Makro, Aufzeichnung beenden aus dem Extras-Menü.



5

Markieren Sie die letzte Zeile des ersten Makros und die ersten Zeilen des neuen Makros bis zum Einfügebefehl.



Ziehen Sie dazu den Mauszeiger mit gedrückter Taste am linken Rand der Zeilen nach unten.

Löschen Sie die Markierung mit **[Entf]** und fügen Sie eine Kommentarzeile ein.

```
Option Explicit
Sub RechnungNeu()
Dim suchname
' RechnungNeu Makro
' Makro sucht nach einem Firmennamen in der Adresskartei.
Sheets("Kundenkartei").Select
Range("A1").Select
Columns("A:A").Select
suchname = InputBox("Bitte geben Sie die Firma ein:", _
    "Adresse suchen")
If suchname = "" Then Exit Sub
Selection.Find(What:=suchname, _
    After:=ActiveCell, _
    LookIn:=xlFormulas, _
    LookAt:=xlPart, _
    SearchOrder:=xlByRows, _
    SearchDirection:=xlNext, _
    MatchCase:=False).Activate
' Neues Blatt einfügen
Sheets.Add
End Sub
```

Lektion

Makroanweisungen für die Rechnung schreiben

Die restlichen Zeilen des Makros, die alle Zellen aus der gesuchten und gefundenen Adresse in das neue Blatt übertragen, können Sie nicht aufzeichnen, hier müssen Sie manuell programmieren. Die wichtigste Anweisung, um Zellen der Tabelle direkt zu beschriften, lautet `Selection.Offset`. Mit dem Objekt `Selection` ist die Markierung gemeint, und der `Offset` bezeichnet den Sprung von dieser Markierung in eine andere Zelle (z.B. `Offset(1,2)` für eine Zeile nach unten und zwei Spalten nach rechts).

Dimensionieren Sie alle erforderlichen Variablen.

```
Option Explicit
Sub RechnungNeu()
Dim suchname
Dim firma As String, anrede As String, vorname As String
Dim nachname As String, strasse As String, plz As String
Dim ort As String
```

Sie können wahlweise eine `Dim`-Anweisung schreiben und mehrere Variablen mit Komma getrennt einführen oder für jede Variable eine eigene `Dim`-Anweisung benutzen. Mit dem Zusatz `as String` bestimmen Sie, dass die Variable nur Texte enthalten darf.

Belegen Sie die Variablen mit den Daten aus der Tabelle.

```

Sub RechnungNeu()
    Dim suchname
    Dim firma As String, anrede As String, vorname As String
    Dim nachname As String, strasse As String, plz As String
    Dim ort As String
    Sheets("Kundenkartei").Select
    Range("A1").Select
    Columns("A:A").Select
    suchname = InputBox("Bitte geben Sie die Firma ein:", _
        "Adresse suchen")
    If suchname = "" Then Exit Sub
    Selection.Find(What:=suchname, _
        After:=ActiveCell, _
        LookIn:=xlFormulas, _
        LookAt:=xlPart, _
        SearchOrder:=xlByRows, _
        SearchDirection:=xlNext, _
        MatchCase:=False).Activate
    ' Adresse in Variablen schreiben
    With ActiveCell
        firma = .Value
        anrede = .Offset(0, 1).Value
        vorname = .Offset(0, 2).Value
        nachname = .Offset(0, 3).Value
        strasse = .Offset(0, 4).Value
        plz = .Offset(0, 5).Value
        ort = .Offset(0, 6).Value
    End With
    ' Neues Blatt einfügen
    Sheets.Add

```

Schreiben Sie die Anweisungen ab `With` nach dem Befehl `Sheets.Add`, vergessen Sie nicht, alles zwischen `With` und `End With` einzurücken.

Die With-Anweisung

Die With-Anweisung ist sehr brauchbar: Da Sie in VBA-Programmen häufig mit Objekten zu tun haben werden und diesen Objekten auch ebenso häufig Eigenschaften und Methoden zuweisen, werden Sie immer wieder lange Objektbezeichnungen einzutippen:

```

Objekt.Eigenschaft1 = Zuweisung
Objekt.Eigenschaft2 = Zuweisung
Objekt.Methode1 = Zuweisung

```

Das Objekt muss jedes Mal wiederholt werden. Die With-Anweisung führt das Objekt einmal ein, und Sie können so lange Eigenschaften und Methoden zuweisen, bis das Objekt nicht mehr gebraucht wird. Mit `End With` beenden Sie die Klammer, wenn das Objekt nicht mehr gebraucht wird oder ein anderes Objekt ins Spiel kommt.

```

With Objekt
    .Eigenschaft1 = Zuweisung
    .Methode1 = Zuweisung
End With

```

3

Schreiben Sie die Daten aus den Variablen in das neue Blatt zurück.

```

' Neues Blatt einfügen
  Sheets.Add
' Adresse in das Rechnungsblatt schreiben
[a1] = firma
[a2] = anrede
[a3] = vorname & " " & nachname
[a4] = strasse
[a6] = plz & " " & ort


[a10] = "Rechnung"
End Sub

```

Jetzt drehen Sie den Spieß um: Die Inhalte der Variablen werden nach der Anweisung `Sheets.Add` in die ersten Zellen der neuen Tabelle geschrieben. Mit der letzten Anweisung fügen Sie noch die Überschrift `Rechnung` in Zelle `A10` ein.

4

Testen Sie das Makro und speichern Sie die Mappe ab.



Sie können gleich anschließend weitere Befehle aufzeichnen und in Ihr Makro integrieren. Alle Formatierungen, ob Fettdruck, Linien, Muster oder Rahmen lassen sich per Makrorekorder in VBA-Befehle umwandeln. Das gleiche gilt für Formeln in den Zellen, die vom Rekorder zwar englischsprachig aufgezeichnet, aber beim Ablauf des Makros korrekt in Deutsch wieder eingetragen werden.

Hier das komplette Listing des Rechnungsmakros noch einmal zur Überprüfung:

```

Option Explicit
Sub RechnungNeu()
  Dim suchname
  Dim firma As String, anrede As String, vorname As String
  Dim nachname As String, strasse As String, plz As String
  Dim ort As String
  Sheets("Kundenkartei").Select
  Range("A1").Select

```

WIE BITTE?

Die Eigenschaft
.value ist die
Standardeigenschaft
einer Zelle. Sie können
sie weglassen, wenn
Sie eine Zelle beschrif-
ten: [a1].value =
firma oder [a1] =
firma.

```
Columns("A:A").Select
suchname = InputBox("Bitte geben Sie die Firma ein:", _
    "Adresse suchen")
If suchname = "" Then Exit Sub
Selection.Find(What:=suchname, _
    After:=ActiveCell, _
    LookIn:=xlFormulas, _
    LookAt:=xlPart, _
    searchOrder:=xlByRows, _
    SearchDirection:=xlNext, _
    MatchCase:=False).Activate
' Adresse in Variablen schreiben
With ActiveCell
    firma = .Value
    anrede = .Offset(0, 1).Value
    vorname = .Offset(0, 2).Value
    nachname = .Offset(0, 3).Value
    strasse = .Offset(0, 4).Value
    plz = .Offset(0, 5).Value
    ort = .Offset(0, 6).Value
End With
' Neues Blatt einfügen
Sheets.Add
' Adresse in das Rechnungsblatt schreiben
[a1].value = firma
[a2].value = anrede
[a3].value = vorname & " " & nachname
[a4].value = strasse
[a6].value = plz & " " & ort

[a10] = "Rechnung"
End Sub
```

Lektion

Der Makrorekorder in anderen Applikationen

Neben Excel bieten noch Word, PowerPoint und Project die Möglichkeit, Aktivitäten im Programmfenster mit dem Makrorekorder aufzuzeichnen und als VBA-Befehle weiter zu bearbeiten. Das Datenbankprogramm Access bietet diesen Komfort leider nicht, ebenso wenig wie die übrigen VBA-fähigen Programme aus der Office-Umgebung (FrontPage, Outlook).

Im Prinzip funktioniert der Makrorekorder in Word genauso wie der Excel-Aufzeichner, nur mit einer gravierenden Einschränkung: Mausaktionen direkt im Dokument oder in der Vorlage können nicht aufgezeichnet werden, nur Symbolleisten- und Menüaufrufe sowie Tastatureingaben. Behelfen Sie sich damit, Aktionen direkt am Text mit Tastenkombinationen auszuführen, der Rekorder zeichnet die passenden Befehle auf.

Suchen Sie in der Hilfe nach dem Suchwort »Tastenkombination«. Hier sind alle wichtigen »Shortcuts« aufgelistet.

Tastenkombination	Erweiterung der Markierung
UMSCHALT+NACH-RECHTS	Um ein Zeichen nach rechts
UMSCHALT+NACH-LINKS	Um ein Zeichen nach links
STRG+UMSCHALT+NACH-RECHTS	Bis zum Wortende
STRG+UMSCHALT+NACH-LINKS	Bis zum Wortanfang
UMSCHALT+ENDE	Bis zum Zeilenende
UMSCHALT+POS1	Bis zum Zeilenanfang
UMSCHALT+NACH-UNTEN	Um eine Zeile nach unten
UMSCHALT+NACH-OBEN	Um eine Zeile nach oben
STRG+UMSCHALT+NACH-UNTEN	Bis zum Absatzende
STRG+UMSCHALT+NACH-OBEN	Bis zum Absatzanfang
UMSCHALT+BILD-AB	Um eine Bildschirmseite nach unten
UMSCHALT+BILD-AUF	Um eine Bildschirmseite nach oben
STRG+UMSCHALT+POS1	Bis zum Dokumentenanfang
STRG+UMSCHALT+ENDE	Bis zum Dokumentenende
ALT+STRG+UMSCHALT+BILD-AB	Bis zum Ende eines Fensters
STRG+A	Um das gesamte Dokument


Üben Sie die Anwendung des Makrorekorders in Word an einem praktischen Beispiel:

Eine Dokumentvorlage präsentiert eine Mitgliederkarte, die per Makro auszufüllen ist. Der Benutzer öffnet ein Dokument mit der Vorlage als Basis und wird sofort mit InputBox-Dialogen aufgefordert, die Daten einzugeben. Damit die Informationen an die richtigen Positionen im Text geschrieben werden können, fügen Sie mit *Einfügen*, *Textmarke* Textmarken ein.

1

Starten Sie Word, erstellen Sie eine neue Vorlage und geben Sie die Beschriftungstexte ein.

Datei, Neu, Vorlage



TIPP


Sie können wahlweise mit Tabellen oder mit Tabulatoren arbeiten.

2

Fügen Sie drei Textmarken mit *Einfügen, Textmarke* in den Text ein.


TIPP

Unter *Extras, Optionen* finden Sie auf der Registerkarte *Ansicht* die Möglichkeit, die Textmarken sichtbar zu machen.




3

Starten Sie den Makrorecorder mit **Alt + F11** oder mit dem Menü.



4

Geben Sie einen Makronamen ein und wählen Sie die neue Vorlage als Speicherort für das Makro.



5

Zeichnen Sie ein Makro *KarteNeu* auf, das den Textanfang, dann die ersten drei Textmarken ansteuert und Daten einträgt.




So bewegen Sie den Cursor im Text:

Tastenkombination	Auswirkung
[Pos1], [Strg] + [Pos1]	Sprung zum Zeilenanfang und Textanfang
[F5]	Sprung zu ... (Textmarke auswählen)
[Esc]	Suche schließen, Text eintragen, Suche wieder starten usw.
[Ende], [Strg] + [Ende]	Sprung zum Zeilenende und Dokumentende

6

Schließen Sie den Makrorekorder und aktivieren Sie mit [Alt] + [F11] den Visual Basic-Editor von Word.



7

Löschen Sie alle Anweisungen, die Sie nicht brauchen.

```
Option Explicit
Sub KarteNeu()
  ' KarteNeu Makro
  Selection.HomeKey Unit:=wdStory
  Selection.GoTo What:=wdGoToBookmark, Name:="vorname"
  Selection.TypeText Text:="Hugo"
  Selection.GoTo What:=wdGoToBookmark, Name:="name"
  Selection.TypeText Text:="Habicht"
  Selection.GoTo What:=wdGoToBookmark, Name:="gdatum"
  Selection.TypeText Text:="1.1.70"
  Selection.EndKey Unit:=wdStory
End Sub
```

Der Makrorekorder zeichnet häufig zu viel auf. Entfernen Sie überflüssige Zeilen einfach mit der Taste **Entf**.

8

Fordern Sie mit der Anweisung **InputBox** die Daten an.

```
Option Explicit
Sub KarteNeu()
  ' KarteNeu Makro
  Dim vname, nname, gdatum, alter
  vname = InputBox("Bitte Vorname eingeben:", "Daten erfassen")
  nname = InputBox("Bitte Nachname eingeben:", "Daten erfassen")
  gdatum = InputBox("Bitte Geburtsdatum eingeben:", "Daten erfassen")
```


Die Variablen werden natürlich sauber dimensioniert.

9

Ändern Sie den aufgezeichneten Code, sodass die Variableninhalte auf die Textmarken geschrieben werden.

ACHTUNG

Geben Sie für Input-Box-Variablen keinen Datentyp an!



10

Geben Sie die Anweisungen zur Altersberechnung ein.

Year() = Jahreszahl
eines Datums

Date = Tagesdatum

```

' Alter berechnen
alter = Year(Date) - Year(gdatum)
Selection.GoTo What:=wdGoToBookmark, Name:="alter"
Selection.TypeText Text:=alter


Selection.EndKey Unit:=wdStory
End Sub

```

Das berechnete Alter wird gleich anschließend auf die vorbereitete Textmarke »alter« geschrieben.


11

Wechseln Sie in das Word-Fenster und aktivieren Sie die Liste mit den aktiven Makros.



12


Starten Sie das neue Makro und testen Sie es.



Das Makro fordert die Daten per InputBox an, berechnet das Alter aus dem Geburtsdatum und setzt die Daten in das Dokument.

ACHTUNG

Alte Daten bitte vorher löschen. Achten Sie darauf, dass die Textmarken nicht entfernt werden dürfen!



Fragen

Testen Sie Ihr Wissen!

- Wie starten Sie die Aufzeichnung eines Makros?
- Wo können Sie kontrollieren, ob der Makrorekorder aktiv ist?
- Nennen Sie zwei Möglichkeiten, um die Makroaufzeichnung zu beenden.
- Mit welcher Anweisung schalten Sie in Excel auf die nächste Zeile um?
- Vollenden Sie dieses Makro:

```
Sub TabINFO()  
With activesheet  
    MsgBox .name  
    MsgBox .cells.count
```
- Schreiben Sie den Inhalt der beiden Variablen »Vorname« und »Nachname« mit einem Leerzeichen dazwischen in die Zelle A1.
- Wie fügen Sie ein neues Tabellenblatt in die Excel-Arbeitsmappe BUDGET.XLS ein?
- Steuern Sie mit einer Anweisung die Word-Textmarke kopf an.
- Wie setzen Sie in der Makroaufzeichnung in Word den Cursor an den Beginn der Zeile?

